
Grenzen regulärer Sprachen
Was reguläre Ausdrücke und endliche Automaten beschreiben können:

lexikale Elemente in Programmiersprachen: Identifier, Numerale, . . .
Morphologie von Worten in natürlichen Sprachen: korrekte Fall-/Zahl-/Zeitendungen, . . .
Systeme, die nur ein endliches Gedächtnis besitzen

Was sie nicht beschreiben können:
geklammerte Ausdrücke in Programmiersprachen
geschachtelte Programmstrukturen wie
if ... while ... if ... endif ... endwhile ... endif
Schachtelung von Haupt- und Nebensätzen in natürlichen Sprachen
Systeme mit einem (potentiell) unendlichen Speicher

Sprache der wohlgeklammerten Ausdrücke ist nicht regulär
{(), (()), ()(), ((())), (())(), ()()(), . . . }

{ anbn | n ≥ 0 } ist nicht regulär
{ε, ab, aabb, aaabbb, aaaabbbb, aaaaabbbbb, . . . }

Kontextfreie Grammatiken

Menge von Ersetzungsregeln
2 Arten von Symbolen: Terminale, Nonterminale
Ausgehend vom Start-Nonterminal werden Nonterminale solange ersetzt, bis nur noch
Terminale bleiben.

Satz → HwP ZwP
HwP → Art Hw
ZwP → Hzw HwP Zw

Art → der | den
Hw → Hund | Mond
Hzw → wird
Zw → anbellen

Satz ⇒p HwP ZwP
⇒p Art Hw Hzw HwP Zw
⇒p der Hund wird Art Hw anbellen
⇒p der Hund wird den Mond anbellen

Generative Grammatiken gehen zurück auf Noam Chomsky (Linguist, Philosoph, Aktivist;
*1928).

Kontextfreie Grammatik
. . . wird beschrieben durch ein 4-Tupel G = ⟨V , T , P, S⟩, wobei

V . . . Menge der Nonterminalsymbole (Variablen)
T . . . Menge der Terminalsymbole (V ∩ T = {})
P ⊆ V × (V ∪ T )∗ . . . Produktionen
S ∈ V . . . Startsymbol

Schreibweise: A → w statt (A, w) ∈ P
A → w1 | · · · | wn statt A → w1, . . . , A → wn

Ableitbarkeit
. . . in einem Schritt:
x A y ⇒ x w y , falls A → w ∈ P und x , y ∈ (V ∪ T )∗.

. . . in mehreren Schritten:
u ∗⇒ v , falls

u = v , oder
u ⇒ u′ und u′ ∗⇒ v für ein Wort u′ ∈ (V ∪ T )∗.

Von Grammatik G generierte Sprache
L(G) = { w ∈ T ∗ | S ∗⇒ w }

Grammatiken G1 und G2 heißen äquivalent, wenn L(G1) = L(G2) gilt.

G = ⟨{S}, {a, b}, {S → ε | a S b}, S⟩

S ∗⇒ aabb, weil S ⇒ aS b ⇒ aa S bb ⇒ aa ε bb = aabb

L(G) = { anbn | n ≥ 0 } = {ε, ab, aabb, aaabbb, . . . }

Kontextfreie Sprachen
Eine Sprache heißt kontextfrei, wenn es eine kontextfreie Grammatik gibt, die sie generiert.

Verschiedene Ableitbarkeitsbegriffe

Linksableitbarkeit: x A y ⇒L x w y falls A → w ∈ P und x ∈ T ∗

(In jedem Schritt wird die linkeste Variable ersetzt.)
Rechtsableitbarkeit: x A y ⇒R x w y falls A → w ∈ P und y ∈ T ∗

(In jedem Schritt wird die rechteste Variable ersetzt.)
Parallelableitbarkeit: x0 A1 x1 · · · An xn ⇒P x0 w1 x1 · · · wn xn
falls A1 → w1, . . . , An → wn ∈ P und x0, . . . , xn ∈ T ∗

(In jedem Schritt werden alle Variablen ersetzt.)
∗⇒L, ∗⇒R und ∗⇒P sind eingeschränkte Ableitungsrelationen:

Manche Wörter w ∈ (V ∪ T )∗ sind mit ∗⇒ herleitbar (S ∗⇒ w),
aber nicht mit diesen Relationen.
Sie können aber jedes Wort der Sprache ableiten. Für alle w ∈ T ∗ gilt:
S ∗⇒ w gdw. S ∗⇒L w gdw. S ∗⇒R w gdw. S ∗⇒P w

L(G) = { w ∈ T ∗ | S ∗⇒ w } = { w ∈ T ∗ | S ∗⇒L w }
= { w ∈ T ∗ | S ∗⇒R w } = { w ∈ T ∗ | S ∗⇒P w }

Kontextfreie Grammatiken - Beispiele

Syntax aussagenlogischer Formeln
G = ⟨{Fm, Op, Var}, T , P, Fm⟩, wobei

T = {⊤, ⊥, ¬, (, ), ∧, ↑, ∨, ↓, ⇔, ⇎, ⇒, ⇐} ∪ V

P = { Fm → Var | ⊤ | ⊥ | ¬Fm | (Fm Op Fm ) ,
Op → ∧ | ↑ | ∨ | ↓ | ⇔ | ⇎ | ⇒ | ⇐ ,
Var → A | B | C | · · · }

((A ↑ B) ↑ (A ↑ B)) ist eine aussagenlogische Formel, weil:

Fm ⇒L (Fm Op Fm ) ⇒L ( (Fm Op Fm )Op Fm )
⇒L ( (Var Op Fm )Op Fm ) ⇒L ( ( AOp Fm )Op Fm )
⇒L ( ( A ↑ Fm )Op Fm ) ⇒L ( ( A ↑ Var )Op Fm )
⇒L ( ( A ↑ B )Op Fm ) ⇒L ( ( A ↑ B ) ↑ Fm )
⇒L ( ( A ↑ B ) ↑ (Fm Op Fm ) ) ⇒L ( ( A ↑ B ) ↑ (Var Op Fm ) )
⇒L ( ( A ↑ B ) ↑ ( AOp Fm ) ) ⇒L ( ( A ↑ B ) ↑ ( A ↑ Fm ) )
⇒L ( ( A ↑ B ) ↑ ( A ↑ Var ) ) ⇒L ( ( A ↑ B ) ↑ ( A ↑ B ) )

9



Kontextfreie Grammatiken - Beispiele

Kontextfreie Grammatiken - Beispiele

Reelle Numerale
G = ⟨V , T , P, Real⟩, wobei

V = {Real , Scale, Sign, Digits, Digit}
T = {0, . . . , 9, ., E, +, -}
und P folgende Produktionen sind:

Real → Digit Digits .Digits Scale
Scale → ε | E Sign Digit Digits
Sign → ε | + | -

Digits → ε | Digit Digits
Digit → 0 | · · · | 9

Optionalität:
A → ε | B (A steht für optionales B)
Wiederholung:
A → ε | B A (A steht für wiederholtes B, auch 0 Mal)
A → B | B A (A steht für wiederholtes B, mind. 1 Mal)

Kontextfreie Grammatiken - Beispiele

Wohlgeformte Klammerausdrücke
WKA ist die kleinste Menge, sodass

ε ∈ WKA
(w) ∈ WKA, wenn w ∈ WKA
w1w2 ∈ WKA, wenn w1, w2 ∈ WKA

G = ⟨ {W }, {(, )}, {W → ε | (W ) | W W }, W ⟩

Beispiel einer Ableitung: W ⇒P W W
⇒P (W )(W )
⇒P ()(W W )
⇒P ()((W )(W ))
⇒P ()(()())

L(G) = {ε, (), (()), ()(), ((())), (())(), ()(()), ()()(), . . . }
= WKA

Induktive Definition vs. kontextfreie Grammatik

Induktive Definition für M kontextfreie Grammatik für M

Mengen M, M0, A1, B1, . . . Nonterminale M, M0, A1, B1, . . .

M ist die kleinste Menge, sodass: M = L(⟨. . . , P, M⟩), wobei P:
M0 ⊆ M M → M0

f (x1, . . . , xm) ∈ M, falls x1 ∈ A1, . . . , xm ∈ Am M → f (A1, . . . , Am)
g(x1, . . . , xn) ∈ M, falls x1 ∈ B1, . . . , xn ∈ Bn M → g(B1, . . . , Bn)
· · · ∈ M, falls . . . M → . . .

h(x1, x2) ∈ M, falls x1, x2 ∈ M M → h(M, M)
h(x , x) ∈ M, falls x ∈ M keine Entsprechung, nicht kontextfrei

Backus-Naur-Form (BNF)

Synonym für kontextfreie Produktionen

Historische Notation:
Nonterminale in spitzen Klammern
Terminale unter Anführungszeichen
::= als Trennsymbol

Durch diese Konvention entfällt die Notwendigkeit, die Mengen der Nonterminale (V ) und der
Terminale (T ) anzugeben.

⟨Real⟩ ::= ⟨Digit⟩ ⟨Digits⟩ "." ⟨Digits⟩ ⟨Scale⟩
⟨Digit⟩ ::= "0" | · · · | "9"

⟨Digits⟩ ::= ε | ⟨Digit⟩ ⟨Digits⟩
⟨Scale⟩ ::= ε | "E" ⟨Sign⟩ ⟨Digit⟩ ⟨Digits⟩
⟨Sign⟩ ::= ε | "+" | "-"

Erweiterte Backus-Naur-Form (EBNF)

„Regular Right Part Grammar“ (Rrpg)
erlaubt reguläre Ausdrücke auf der rechten Seite von Produktionen
Bessere Lesbarkeit durch Vermeidung von Rekursionen und Leerwörtern
Kompaktere Spezifikationen
keine echte Erweiterung: EBNF-Notationen lassen sich eliminieren

Elimination der EBNF-Notation:

A → w1 (w2) w3 entspricht A → w1 B w3
B → w2

A → w1 {w2} w3 entspricht A → w1 B w3
B → ε | w2 B

A → w1 [w2] w3 entspricht A → w1 B w3
B → ε | w2

16



Listen von Bezeichnern
Ebnf: IdList → Id { "," Id }

Ohne Ebnf: IdList → Id B
B → ε | "," Id B

oder IdList → Id | Id "," IdList

Skalierungsfaktor der reellen Numerale
Ebnf: Scale → "E" ["+" | "-"] Digit {Digit}

Ohne Ebnf: Scale → "E"B1 Digit B2
B1 → ε | "+" | "-"
B2 → ε | Digit B2

Syntaxdiagramme

Mit rekursiven Diagrammen können beliebige kontextfreie Grammatiken in Ebnf dargestellt
werden.

Wohlgeklammerte Ausdrücke

W =
( W )

W W

W → (W )

| ε

| W W

18



Reguläre und Kontextfreie Sprachen

Reguläre Sprachen
Kontextfreie Grammatiken

Motivation & Definition
Beispiele
Andere Notationen
Style Guide für Grammatiken
Grammatiken in freier Wildbahn
Jenseits der Kontextfreiheit

19



Style Guide für Grammatiken

Wiederholungsoperator statt Rekursion (wenn möglich).

Schlecht: A → a | A A oder A → a | a A

Besser: a { a } oder a + oder . . . (je nach gewählter Notation)

Optionsoperator statt Leerwort.

Schlecht: ε | A

Besser: [ A ] oder A ? oder . . . (je nach gewählter Notation)

Modularisierung statt Duplizierung.

Schlecht: Identifier → ("a" | · · · | "z") { "a" | · · · | "z" | "0" | · · · | "9" }

Besser: Identifier → Letter { Letter | Digit }
Letter → "a" | · · · | "z"
Digit → "0" | · · · | "9"

Verwendung sprechender Namen.

Schlecht: X → Y { Y | Z }
Y → "a" | · · · | "z"
Z → "0" | · · · | "9"

Besser: Siehe oben.

Klare Unterscheidung zwischen Terminalen, Nonterminalen und Metanotationen.

Schlecht: \begin{tabular}[[Position]]{Spalte{Spalte}}

Besser: "\begin{tabular}" [ "[" Position "]" ] "{" Spalte { Spalte } "}"

"\begin{tabular}" [ "[" ⟨Position⟩ "]" ] "{" ⟨Spalte⟩ { ⟨Spalte⟩ } "}"

Inhaltliche Strukturierung statt Minimierung der Nonterminale und Regeln.

Schlecht:
"\begin{tabular}" [ "[" ("t" | "b") "]" ] "{" ("l" | "c" | "r") { "l" | "c" | "r" } "}"

Besser: . . . → "\begin{tabular}" [ Position ] Spalten
Position → "[b]" | "[t]"
Spalten → "{" Spalte { Spalte } "}"
Spalte → "l" | "c" | "r"

Beispiel: Tabellen in LaTeX

TeX . . . Textsatzsystem von Donald E. Knuth
LaTeX . . . TeX-Makros für „logisches Markup“ von Leslie Lamport

\begin{tabular}[t]{lc}
Eintrag 11 & Eintrag 12 \\
Eintrag 21 &

\begin{tabular}{rr}
Eintrag 22 & ist selber \\
eine & Tabelle.
\end{tabular}\\

Eintrag 31 & Eintrag 32
\end{tabular}

Eintrag 11 Eintrag 12

Eintrag 21 Eintrag 22 ist selber
eine Tabelle.

Eintrag 31 Eintrag 32

Gesucht: Kontextfreie Grammatik G in Ebnf für die Sprache der LaTeX-Tabellen

Beispiel: Tabellen in LaTeX
G = ⟨V , T , P, Tabelle⟩, wobei:

P = { Tabelle → "\begin{tabular}" [ Position ] Spalten
Zeilen
"\end{tabular}" ,

Position → "[b]" | "[t]" ,
Spalten → "{" Spalte { Spalte } "}" ,
Spalte → "l" | "c" | "r" ,

Zeilen → Zeile { "\\" Zeile } ,
Zeile → Eintrag { "&" Eintrag } ,

Eintrag → { Tabelle | Zeichen } ,
Zeichen → "0" | · · · | "9" | "a" | · · · | "Z" | "." | "␣" }

V = {Tabelle, Position, Spalten, Spalte, Zeilen, Zeile, Eintrag , Zeichen}
T = { "0", . . . , "9", "a", . . . , "z", "A", . . . , "Z",

".", "␣", "{", "}", "[", "]", "&", "\" }

Nur eine Rekursion für Schachtelung der Tabellen notwendig!

N. Wirth: Programming in Modula-2

J. Gosling et al.: Java Language Specification

591

C H A P T E R 18
Syntax

THIS chapter presents a grammar for the Java programming language.

The grammar presented piecemeal in the preceding chapters (§2.3) is much better
for exposition, but it is not well suited as a basis for a parser. The grammar presented
in this chapter is the basis for the reference implementation. Note that it is not an
LL(1) grammar, though in many cases it minimizes the necessary look ahead.

The grammar below uses the following BNF-style conventions:

• [x] denotes zero or one occurrences of x.

• {x} denotes zero or more occurrences of x.

• x | y means one of either x or y.

Identifier:
    IDENTIFIER

QualifiedIdentifier:
    Identifier { . Identifier }

QualifiedIdentifierList: 
    QualifiedIdentifier { , QualifiedIdentifier }

SYNTAX

606

EnumBody:
    { [EnumConstants] [,] [EnumBodyDeclarations] }

EnumConstants:
    EnumConstant
    EnumConstants , EnumConstant

EnumConstant:
    [Annotations] Identifier [Arguments] [ClassBody]

EnumBodyDeclarations:
    ; {ClassBodyDeclaration}

AnnotationTypeBody:
    { [AnnotationTypeElementDeclarations] }

AnnotationTypeElementDeclarations:
    AnnotationTypeElementDeclaration
    AnnotationTypeElementDeclarations AnnotationTypeElementDeclaration

AnnotationTypeElementDeclaration:
    {Modifier} AnnotationTypeElementRest

AnnotationTypeElementRest:
    Type Identifier AnnotationMethodOrConstantRest ;
    ClassDeclaration
    InterfaceDeclaration
    EnumDeclaration  
    AnnotationTypeDeclaration

AnnotationMethodOrConstantRest:
    AnnotationMethodRest
    ConstantDeclaratorsRest  

AnnotationMethodRest:
    ( ) [[]] [default ElementValue]

Jenseits der Kontextfreiheit

Formale Sprachen
{ anbncn | n ≥ 0 } ist nicht kontextfrei.

Formale Sprachen
{ ww | w ∈ {a, b, c}∗ } ist nicht kontextfrei.

Programmiersprachen
Typen- und Deklarationsbedingungen nicht oder nur schwer kontextfrei darstellbar.

Natürliche Sprachen (Schweizer Dialekt)
Mer d’chind em Hans es huus
haend wele laa hälfe aastriiche.

Wir wollten die Kinder dem Hans
helfen lassen das Haus anzustreichen.

Kontextfreie Sprachen & Kellerautomaten

Reguläre Sprache besitzen die gleich Ausdrucksstärke wie DEAs und NEAs.
Gibt es auch einen passenden Automaten für Kontextfreie Sprachen?
↪→ Ja, nichtdeterministische Kellerautomaten (pushdown automata)
Kellerautomat = endlicher Automat + Kellerspeicher (Stack)

Kellerspeicher ermöglicht unendlich viele “Zustände”
Automat sieht nur das oberste Symbol im Keller (und Zustandsübergänge hängen davon ab)
Bei einem Übergang können neue Symbole oben auf den Keller geschrieben werden
oder das oberste Symbol kann gelöscht werden.

Andere Arten von Grammatiken

Wir haben nur kontextfreie Grammatiken (Typ-2 Grammatik) betrachtet

Es gibt ausdrucksstärkere Grammatiken:
uneingeschränkte formale Grammatik (Typ-0)
kontextsensitive Grammatik (Typ-1)

. . . und einfachere Arten von Grammatiken:
reguläre Grammatiken (Typ-3) definieren die regulären Sprachen und besitzen damit dieselbe
Ausdrucksstärke wie endliche Automaten und reguläre Ausdrücke.

Zusammenfassung

Grammatik definieren formale Sprachen

Kontextfreie Grammatiken
ausdrucksstärker als reguläre Ausdrücke und endliche Automaten
darstellbar als rekursive Syntaxdiagramme

Style Guide für Grammatiken
EBNF kombiniert reguläre Ausdrücke und kontextfreie Produktionen

